
1.1: Introduction

In many programming languages, there is a built-in or widely-distributed data type for storing time. But in every instance where there is a Time data type, it is invariably a time of day and inseparably associated with date. Even a time entered without a date, like 2:45:00, is automatically assumed to be 2:45 AM on January 1st 2000 (or some other default date). No programming language available today has the ability to store, as its own data type, time independent of date or day – elapsed time. Yet the capability to remember an elapsed time is consistently needed in numerous applications: a business using a high-traffic mainframe wants to know how long certain programs take to run to see if they are losing valuable computer time to inefficient software, or a marathon runner wants to keep track of race times and calculate pace charts. Most of the time the people who need to use elapsed time find some way to work around it using existing data types – subtracting one date/time from another if they are on the same day, or storing hours, minutes, and seconds in three separate numeric fields. But most such workarounds result in either a reduction in functionality or an increase in the complexity of the code, and are inadequate compromises.

The Tremel Elapsed Time Standard aims to solve this major problem by designing and implementing a standard data type for elapsed time. While there may already be “unofficial” Elapsed Time data types out there, designed by experienced programmers who needed elapsed-time functionality and had the capability to modify their programming environment, there is no way of guaranteeing that one programmer’s invented data type will be compatible with another’s system. This standard will present a method of formatting Elapsed Time data that can be easily followed for simple cross-system standardization. There are probably better ways of doing it already invented, and the best way to make an Elapsed Time data type will probably continue to evolve, so this is an open-source standard that may be reviewed and modified by anyone – as long as the modifications are submitted and recorded in the standard document so that a person can easily know what the most current version is. Sections 1.4 and 1.5 contain the Visual Basic code used for Version 1.0 of the standard, and this may be copied or distributed freely as long as the reference to the Tremel Elapsed Time Standard is maintained.
1.2: Using the Input Function

Currently the Tremel Elapsed Time Standard does not use a custom data type (i.e. written using a class module), but simulates the functionality of an Elapsed Time data type using existing types and a set of functions. The input function, called TotalSeconds in the standard distribution, receives literal elapsed time as a string and returns that elapsed time as a number of seconds. The input string must be in hours-minutes-seconds format, but the three numbers can be separated by any non-numeric character other than a period (which is interpreted to mean a decimal) and hours can be omitted if the time is only minutes and seconds. Since there is no way to input days, hours can be greater than 24 without causing an error. Thus the following are all valid inputs:
· “2:34:00” (2 hours 34 minutes)

· “2-34” (2 minutes 34 seconds)

· “2 34 30” (2 hours 34 minutes 30 seconds)

· “2*34.5” (2 minutes 34½ seconds)

· “26_34:05” (1 day 2 hours 34 minutes 5 seconds)

The number returned by the function is a double-precision floating-point number, to accommodate the fact that (a) the number of seconds in times spanning several days will be a very large number, and (b) there may be fractional seconds (so a decimal point is needed).

1.3: Using the Output Function

Once an elapsed time is input as a string and stored as a number, it will need to be displayed once again as a string. The output function, called StopwatchTime in the standard distribution, receives a total number of seconds and returns that number of seconds as an elapsed-time string. Although any separator could be used in the input function, the output function always uses colons to separate groups of numbers because that’s the standard format and some arbitrary separator must be chosen to use every time. If the number of seconds passed to the function would result in 00 hours, the hours part of the string is omitted and only minutes and seconds are returned. Minutes will not be omitted if they are zero, however, to avoid confusion between strings returned by the function and numbers used elsewhere in the application (57.5 could be a string or a number, but 00:57.5 would definitely be a string).
1.4: Structure of the Input Function

The input function uses a stack to parse the string it is passed and accumulate the total number of seconds each time it encounters a separator character. It and its supporting functions (the stack-handling ones) are given below. Inline comments have been removed because they are all much wider than the page and cause confusing line breaks. 
Private strStack(1 To 20) As String
Private intTop As Integer
'***********************************************************************

'   TotalSeconds Function

'   By Edward Tremel, 1/21/06

'

'   This function is passed a string of "elapsed time" in the form

'   hours:minutes:seconds (at a maximum; hours, and minutes can be

'   omitted) and returns a double of the total number of seconds in that

'   elapsed-time string. It parses the string, using a stack to store

'   each numeral until it reaches a separator character (any single

'   character except a period), at which point it adds the number

'   collected to the total number of seconds (converting appropriately).

'

'   Input: Elapsed (stopwatch) time as a string

'   Output: The total number of seconds in that time as a double

'***********************************************************************

Public Function TotalSeconds(ByVal strStopwatchTime As String) As Double
On Error GoTo Err_TotalSeconds

    Dim strCurChar As String
    Dim sglNumber As Single     
    Dim dblTotal As Double
    Dim intCounter As Integer
    For intCounter = 1 To Len(strStopwatchTime)

        strCurChar = Mid(strStopwatchTime, intCounter, 1)

        If IsNumeric(strCurChar) = True Or strCurChar = "." Then
            Call PushStack(strCurChar)

        Else    
            sglNumber = PopStack()

            dblTotal = (dblTotal + sglNumber) * 60

        End If
    Next intCounter

    sglNumber = PopStack()

    dblTotal = dblTotal + sglNumber 

    TotalSeconds = dblTotal
Exit_TotalSeconds:

    Exit Function
Err_TotalSeconds:

    MsgBox Err.Description, , "Error " + Err.Number

    Resume Exit_TotalSeconds
End Function
'***********************************************************************

'   PushStack Subroutine

'   By Edward Tremel, 1/30/06

'

'   This is the "push" part of a basic stack storage system, where the

'   array strStack() is used to store the items in the stack. Although

'   a true stack is dynamic, it's much easier to use a fixed-length

'   array in Visual Basic, and as previoiusly noted it's long enough for

'   the stack to seem open-ended in this application. This is a

'   subroutine rather than a function because it doesn't have anything

'   it needs to return.

'

'   Input: A string of the next character to be added to the stack

'   Output: None

'***********************************************************************

Public Sub PushStack(ByVal strNewValue As String)
On Error GoTo Err_PushStack

    intTop = intTop + 1

    strStack(intTop) = strNewValue
Exit_PushStack:

    Exit Sub

Err_PushStack:

    MsgBox Err.Description, , "Error " + Err.Number

    Resume Exit_PushStack

End Sub
'***********************************************************************

'   PopStack Function

'   By Edward Tremel, 1/30/06

'

'   This is the "pop" part of a basic stack storage system, with one

'   major difference: this function always pops the whole stack, rather

'   than a limited number of values. It's a custom "pop" function used

'   only in the Tremel Elapsed Time Standard functions, and within this

'   use will never need to pop any fewer values than the entire stack.

'   Similarly, within the function it combines the popped numerals into

'   a single variable, because the only thing the stack system is used

'   for within this set of functions is to store numbers one digit at a

'   time. The recombining of digits is done within this function rather

'   than within the function calling it to preserve simplicity (since in

'   the calling function it would be performing the same operations every

'   time it called the PopStack function).

'

'   Input: None

'   Output: The number that was stored in the stack as parsed digits,

'           as a single (to accomodate fractional seconds).

'***********************************************************************

Public Function PopStack() As Single
On Error GoTo Err_PopStack

    Dim sglNumber As Single
    Dim intCounter As Integer
    Dim intExponent As Integer
    Dim strCurElement As String
    Dim intCurElement As Integer
    For intCounter = intTop To 1 Step -1

        strCurElement = strStack(intCounter)

        If strCurElement = "." Then  
            sglNumber = sglNumber / (10 ^ intExponent)  
            intExponent = 0                             

        Else
            intCurElement = strCurElement

            sglNumber = sglNumber + (intCurElement * (10 ^ intExponent))

            intExponent = intExponent + 1   
        End If                              
    Next intCounter

    intTop = 0

    PopStack = sglNumber
Exit_PopStack:

    Exit Function
Err_PopStack:

    MsgBox Err.Description, , "Error " + Err.Number

    Resume Exit_PopStack
End Function
1.5: Structure of the Output Function

The output function divides out the total number of seconds into hours, minutes, and seconds, then builds the elapsed-time string to return. Fractional seconds are rounded to 10 decimal places (both by adding .0000000001 to the minutes before taking the integer part and by using the Round function on the seconds) to eliminate precision-caused errors. That is, the rounding is done to ensure that, for example, .666666666666666 - .333333333333333 results in 0 and not .0000000000000001. Again, inline comments have been removed from the code to reduce visual clutter, since this page is narrower than the screen in the code editor.

'***********************************************************************

'   StopwatchTime Function

'   By Edward Tremel, 1/21/06

'

'   This function is passed a number of seconds (a double) and returns a

'   string of that many seconds in hours:minutes:seconds format. If

'   hours are 0, it omits them, but it will always print at least

'   minutes and seconds. Regardless of the separator used in the input

'   function (by the user), it will always use colons as separators in

'   the output.

'

'   Input: A number of seconds as a double (partial seconds can be used)

'   Output: Elapsed (stopwatch) time as a string

'***********************************************************************

Public Function StopwatchTime(ByVal dblTotalSeconds As Double) As String
On Error GoTo Err_StopwatchTime

    Dim intHours As Integer
    Dim intMinutes As Integer
    Dim sglSeconds As Single
    Dim strTimeString As String
    Dim dblHoursWithDecimal As Double
    Dim strMinutes As String
    Dim strSeconds As String
    Const sglTolerance As Single = 0.0000000001 

    dblHoursWithDecimal = dblTotalSeconds / 60 / 60

    intHours = Int(dblHoursWithDecimal)

    intMinutes = Int((dblHoursWithDecimal - intHours) * 60 + sglTolerance)

    sglSeconds = Round((((dblHoursWithDecimal - intHours) * 60) - _

    intMinutes) * 60, 10) 

    If intMinutes < 10 And intHours > 0 Then                                            
        strMinutes = "0" & intMinutes

    Else                                
        strMinutes = intMinutes         

    End If                              

    If sglSeconds < 10 Then             
        strSeconds = "0" & sglSeconds   

    Else                                
        strSeconds = sglSeconds         

    End If
    If intHours = 0 Then
        strTimeString = strMinutes & ":" & strSeconds

    Else
        strTimeString = intHours & ":" & strMinutes & ":" & strSeconds

    End If
    StopwatchTime = strTimeString
Exit_StopwatchTime:

    Exit Function
Err_StopwatchTime:

    MsgBox Err.Description, , "Error " + Err.Number

    Resume Exit_StopwatchTime
End Function
